OEIS A129392

Richard J. Mathar[*]
Leiden Observatory, P.O. Box 9513, 2300 RA Leiden, The Netherlands
(Dated: October 14, 2008)

An aid to construction of sequences A129392, A129394, and A129396 of the Online Encyclopedia of Integer Sequences is provided here.

A129392 defines a symmetric triangle

$$T(r,c) = T(r,2r-c) = \begin{pmatrix} 1 \\ 1 & 3 & 1 \\ 1 & -5 & -7 & -5 & 1 \\ 1 & 7 & 5 & 3 & 5 & 7 & 1 \\ 1 & 3 & 1 & -9 & -15 & -9 & 1 & 3 & 1 \\ 1 & \vdots & \vdots & \vdots & \vdots & \ddots & \ddots & \ldots & 1 \end{pmatrix}; \quad 1 \le c \le 2r+1; \quad 1 \le r. \tag{1}$$

where the non-redundant part, the first r columns in row r up to the diagonal, is

$$T(r,c) = \begin{pmatrix} 1 \\ 1 & 3 \\ 1 & -5 & -7 \\ 1 & 7 & 5 & 3 \\ 1 & 3 & 1 & -9 & -15 \\ 1 & -33 & -35 & -5 & 29 & 43 \\ 1 & 91 & 89 & 31 & -23 & -49 & -55 \\ 1 & -137 & -139 & -93 & -43 & 19 & 85 & 115 \\ 1 & 51 & 49 & 135 & 225 & 183 & 1 & -201 & -287 \\ 1 & 399 & 397 & -53 & -499 & -581 & -307 & 91 & 397 & 507 \\ 1 & \vdots & \vdots & \vdots & \vdots & \ddots & \ddots \end{pmatrix} \tag{2}$$

The non-redundant part of the triangle A129394 is

$$T(r,c) = \begin{pmatrix} 0 \\ 1 & 2 \\ 2 & -5 & -8 \\ 3 & 9 & 4 & 0 \\ 4 & 19 & 12 & -16 & -32 \\ 5 & -107 & -116 & -28 & 76 & 120 \\ 6 & 287 & 276 & 96 & -64 & -132 & -144 \\ 7 & -367 & -380 & -300 & -196 & 8 & 268 & 392 \\ 8 & -245 & -260 & 352 & 992 & 940 & 144 & -804 & -1216 \\ 9 & 2733 & 2716 & 228 & -2228 & -2808 & -1572 & 296 & 1724 & 2232 \\ & \vdots & \vdots & \vdots & \vdots & \ddots & \ddots \end{pmatrix} \tag{3}$$

---

[*] URL: http://www.strw.leidenuniv.nl/~mathar

The non-redundant part of the triangle A129396 is

$$T(r,c)=\begin{pmatrix}
1\\
2 & 5\\
3 & -10 & -15\\
4 & 16 & 9 & 3\\
5 & 22 & 13 & -25 & -47\\
6 & -140 & -151 & -33 & 105 & 163\\
7 & 378 & 365 & 127 & -87 & -181 & -199\\
8 & -504 & -519 & -393 & -239 & 27 & 353 & 507\\
9 & -194 & -211 & 487 & 1217 & 1123 & 145 & -1005 & -1503\\
10 & 3132 & 3113 & 175 & -2727 & -3389 & -1879 & 387 & 2121 & 2739\\
 & \vdots & \vdots & \vdots & \vdots & \ddots & \ddots
\end{pmatrix} \tag{4}$$

Supposed the edge element $T(r,1)$ is defined and the previous row $T(r-1,c)$ is known, and the left-right symmetry is made explicit for the element in the middle of the row, the 3-way binding

$$T(r,c)=T(r-1,c-1)+T(r,c-1)+T(r,c+1),\quad r=2,3,4,\ldots;c=2,3,\ldots,2r, \tag{5}$$

may be written as a linear system of $r-1$ equations for the independent elements of row r,

$$\begin{pmatrix}
1 & -1 & 0 & 0 & 0 & 0 & \cdots\\
-1 & 1 & -1 & 0 & 0 & 0 & \cdots\\
0 & -1 & 1 & -1 & 0 & 0 & \cdots\\
0 & 0 & -1 & 1 & -1 & 0 & \cdots\\
\vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \\
0 & 0 & \cdots & -1 & 1 & -1 & 0\\
0 & 0 & \cdots & 0 & -1 & 1 & -1\\
0 & 0 & 0 & \cdots & 0 & -2 & 1
\end{pmatrix}\cdot\begin{pmatrix}T(r,2)\\T(r,3)\\T(r,4)\\\vdots\\T(r,r-1)\\T(r,r)\end{pmatrix}=\begin{pmatrix}T(r-1,1)+T(r,1)\\T(r-1,2)\\T(r-1,3)\\\vdots\\T(r-1,r-1)\end{pmatrix}. \tag{6}$$

with a tri-diagonal matrix of coefficients. *Note that (5) defines column numbers to start at 1 from the left in each row, whereas the current definition in the OEIS starts column enumeration at r.* In particular we have

$$T(r,1)=1;\quad T(r-1,1)+T(r,1)=2;\quad (A129392) \tag{7}$$
$$T(r,1)=r-1;\quad T(r-1,1)+T(r,1)=2r-3;\quad (A129394) \tag{8}$$
$$T(r,1)=r;\quad T(r-1,1)+T(r,1)=2r;\quad (A129396) \tag{9}$$

on the right hand side for $r=1,2,3,\ldots$. If we call the matrix on the left hand side A_{r-1}, index according to number of rows or columns, its determinant is expanded along the first column and first row with the checkerboard rule,

$$\det A_{r-1}=\det\begin{vmatrix}
1 & -1 & 0 & 0 & 0 & 0 & \cdots\\
-1 & 1 & -1 & 0 & 0 & 0 & \cdots\\
0 & -1 & 1 & -1 & 0 & 0 & \cdots\\
0 & 0 & -1 & 1 & -1 & 0 & \cdots\\
\vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \\
0 & 0 & \cdots & -1 & 1 & -1 & 0\\
0 & 0 & \cdots & 0 & -1 & 1 & -1\\
0 & 0 & 0 & \cdots & 0 & -2 & 1
\end{vmatrix}=\det A_{r-2}+\det\begin{vmatrix}
-1 & 0 & 0 & 0 & 0 & \cdots\\
-1 & 1 & -1 & 0 & 0 & \cdots\\
0 & -1 & 1 & -1 & 0 & \cdots\\
\ddots & \ddots & \ddots & \ddots & \ddots & \\
0 & \cdots & -1 & 1 & -1 & 0\\
0 & \cdots & 0 & -1 & 1 & -1\\
0 & 0 & \cdots & 0 & -2 & 1
\end{vmatrix}=\det A_{r-2}-\det A_{r-3}. \tag{10}$$

Initial values are

$$\det A_3=\det\begin{vmatrix}1 & -1 & 0\\-1 & 1 & -1\\0 & -2 & 1\end{vmatrix}=1-(-2)(-1)-(-1)(-1)=-2; \tag{11}$$

$$\det A_4=\det\begin{vmatrix}1 & -1 & 0 & 0\\-1 & 1 & -1 & 0\\0 & -1 & 1 & -1\\0 & 0 & -2 & 1\end{vmatrix}=\det\begin{vmatrix}1 & -1 & 0\\-1 & 1 & -1\\0 & -2 & 1\end{vmatrix}+\det\begin{vmatrix}-1 & 0 & 0\\-1 & 1 & -1\\0 & -2 & 1\end{vmatrix}=\det A_3+(-1)-(-2)(-1)(-1)=-1. \tag{12}$$

With the recursion (10) we get the sequence

$$\det A_{3,4,5,\ldots} = -2, -1, 1, 2, 1, -1, -2, -1, \ldots \tag{13}$$

which is periodic with a period of 6. If we define A_{r-1} more precisely as what is left over if a number of rows and columns have been eliminated from the top and left, we also have

$$\det A_2 = \det \begin{vmatrix} 1 & -1 \\ -2 & 1 \end{vmatrix} = -1; \quad \det A_1 = 1. \tag{14}$$

This sequence of determinants is essentially the same as A057079, A087204, A100051, and A122876. It therefore is never 0 which shows that the solution of (6) is unique. Cramer's rule for $T(r,2)$ in (6) yields

$$T(r,2) = \frac{\det \begin{vmatrix} T(r-1,1)+T(r,1) & -1 & 0 & 0 & 0 & 0 & \cdots \\ T(r-1,2) & 1 & -1 & 0 & 0 & 0 & \cdots \\ T(r-1,3) & -1 & 1 & -1 & 0 & 0 & \cdots \\ T(r-1,4) & 0 & -1 & 1 & -1 & 0 & \cdots \\ \vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \\ T(r-1,r-3) & 0 & \cdots & -1 & 1 & -1 & 0 \\ T(r-1,r-2) & 0 & \cdots & 0 & -1 & 1 & -1 \\ T(r-1,r-1) & 0 & 0 & \cdots & 0 & -2 & 1 \end{vmatrix}}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + \det \begin{vmatrix} T(r-1,2) & -1 & 0 & 0 & 0 & \cdots \\ T(r-1,3) & 1 & -1 & 0 & 0 & \cdots \\ T(r-1,4) & -1 & 1 & -1 & 0 & \cdots \\ \vdots & & \ddots & \ddots & \ddots & \cdots \\ T(r-1,r-2) & \vdots & 0 & -1 & 1 & -1 \\ T(r-1,r-1) & \vdots & 0 & 0 & -2 & 1 \end{vmatrix}}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + T(r-1,2) \det A_{r-3} + \det \begin{vmatrix} T(r-1,3) & -1 & 0 & 0 & \cdots \\ T(r-1,4) & 1 & -1 & 0 & \cdots \\ \vdots & & \ddots & \ddots & \cdots \\ T(r-1,r-2) & 0 & -1 & 1 & -1 \\ T(r-1,r-1) & 0 & 0 & -2 & 1 \end{vmatrix}}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + \sum_{c=2}^{r-5} T(r-1,c) \det A_{r-c-1} + \det \begin{vmatrix} T(r-1,r-4) & -1 & 0 & 0 \\ T(r-1,r-3) & 1 & -1 & 0 \\ T(r-1,r-2) & -1 & 1 & -1 \\ T(r-1,r-1) & 0 & -2 & 1 \end{vmatrix}}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + \sum_{c=2}^{r-4} T(r-1,c) \det A_{r-c-1} + \det \begin{vmatrix} T(r-1,r-3) & -1 & 0 \\ T(r-1,r-2) & 1 & -1 \\ T(r-1,r-1) & -2 & 1 \end{vmatrix}}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + \sum_{c=2}^{r-4} T(r-1,c) \det A_{r-c-1} + T(r-1,r-1) + T(r-1,r-2) - T(r-1,r-3)}{\det A_{r-1}}$$

$$= \frac{[T(r-1,1)+T(r,1)] \det A_{r-2} + \sum_{c=2}^{r-2} T(r-1,c) \det A_{r-c-1} + T(r-1,r-1)}{\det A_{r-1}}, \quad r > 2. \tag{15}$$

This could probably also be derived for more generic expansion of the determinant of lower Hessenberg matrices [1, 2]. Using this equation to calculate $T(.,2)$ in combination with

$$T(r,2) = 2T(r,1) + T(r-1,1), \quad r = 2, \tag{16}$$

the remaining elements in each row then follow from (5),

$$T(r,c+1) = T(r,c) - T(r-1,c-1) - T(r,c-1). \tag{17}$$

This is implemented in the following Maple program.

The row sums of the full tables are

$$\begin{aligned}
A129393 &= 1, 5, -15, 29, -23, -43, 225, -499, 601, 197, -2991, 8189, -12599, 5045, 35265,\\
&\quad -125971, 236857, -206683, -327375, 1808861, -4117079, 5115797, 1120929,\\
&\quad -23825971, 66994201, -105678715, 49059345, 275536829, -1022847863, 1966396277,\\
&\quad -1807797375, -2442192979, 14557768441, -33904533403, 43482526449, 5170554269,\\
&\quad -189441768599, 547643088725, -885162191775, 464914220429, 2145906105817,\\
&\quad -8297375199163, 16308501174225, -15736002726019, -18025996518839\ldots\\
A129395 &= 0, 4, -14, 32, 6, -220, 794, -1528, 1038, 4428, -19998, 44528,\\
&\quad -49770, -48908, 391210, -1033384, 1520286, -160228, -6341134,\\
&\quad 20818496, -37590234, 26378436, 82578234, -374832472, 813526830,\\
&\quad -912940820, -677582846, 6058167440, -15935945034, 23495866004,\\
&\quad -4618822582, -86184783880, 286703024574, -520158934596, 389338165266,\\
&\quad 1004756849888, -4750723870650, 10401903764068, -11992691823782,\\
&\quad -6934949747896, 71851351085646, -191819817252724, 287767798191650\ldots\\
A129397 &= 1, 9, -29, 61, -17, -263, 1019, -2027, 1639, 4625, -22989,\\
&\quad 52717, -62369, -43863, 426475, -1159355, 1757143, -366911, -6668509,\\
&\quad 22627357, -41707313, 31494233, 83699163, -398658443, 880521031,\\
&\quad -1018619535, -628523501, 6333704269, -16958792897, 25462262281,\\
&\quad -6426619957, -88626976859, 301260793015, -554063467999, 432820691715,\\
&\quad 1009927404157, -4940165639249, 10949546852793, -12877854015557, -6470035527467,\\
&\quad 73997257191463, -200117192451887, 304076299365875
\end{aligned} \tag{18}$$

One step of a Gauss elimination in (6) (sum of rows 1 and 2 provides the new row 2) converts this to

$$\begin{pmatrix}
1 & -1 & 0 & 0 & 0 & 0 & \cdots \\
0 & 0 & -1 & 0 & 0 & 0 & \cdots \\
0 & -1 & 1 & -1 & 0 & 0 & \cdots \\
0 & 0 & -1 & 1 & -1 & 0 & \cdots \\
\vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \\
0 & 0 & \cdots & -1 & 1 & -1 & 0 \\
0 & 0 & \cdots & 0 & -1 & 1 & -1 \\
0 & 0 & 0 & \cdots & 0 & -2 & 1
\end{pmatrix}
\cdot
\begin{pmatrix}
T(r,2) \\ T(r,3) \\ T(r,4) \\ \vdots \\ T(r,r-1) \\ T(r,r)
\end{pmatrix}
=
\begin{pmatrix}
T(r-1,1) + T(r,1) \\
T(r,1) + T(r-1,1) + T(r-1,2) \\
T(r-1,3) \\
\vdots \\
T(r-1,r-1)
\end{pmatrix}. \tag{19}$$

Solving the second row gives

$$T(r,4) = -[T(r,1) + T(r-1,1) + T(r-1,2)]. \tag{20}$$

in terms of known elements. (20) is not actually used or needed.

```
# Determinant of a tridiagonal matrix with 1 on the diagonal, -1 on the
# adjacent sub-diagonals, with the exception that one of the two subdiagonals
# has a -2 on the last row.
detA := proc(r)
        local rmod ;
        if r = 0 then
                # determinant of nil matrix is 1 by definition
                1 ;
        else
                # Periodic with period 6. Return 1,-1,-2,-1,2,1,.. for
                # r= 1,2,3,4,...
```

```
                rmod := abs(((r-1) mod 6 ) -2) ;
                if rmod =3  then
                        2 ;
                elif rmod =2 then
                        1 ;
                elif rmod =1 then
                        -1 ;
                else
                        -2 ;
                fi ;
        fi ;
end:

# return the first r columns of row r, given that the
# element of colum 1 is set to c1 and corresponding recursive call for row
# r-1 gave Tr_1.
# See http://www.strw.leidenuniv.nl/~mathar/progs/a129392.pdf
T := proc(c1,Tr_1,r)
        local a, c2 ,c ;
        a := [c1] ;
        if r > 1 then
                if r = 2 then
                        c2 := 2*c1+ op(1,Tr_1) ;
                        a := [op(a),c2] ;
                else
                        c2 := (c1+ op(1,Tr_1))*detA(r-2)+op(-1,Tr_1) ;
                        for c from 2 to r-2 do
                                c2 := c2+ op(c,Tr_1)*detA(r-c-1) ;
                        od ;
                        c2 := c2/detA(r-1) ;
                        a := [op(a),c2] ;
                        for c from 3 to r do
                                c2 := op(-1,a)-op(-2,a)-op(c-2,Tr_1) ;
                                a := [op(a),c2] ;
                        od ;
                fi ;
        fi ;
        RETURN(a) ;
end:

# return list of the first r elements of row r
A129392 := proc(r) option remember ;
        local Tprev ;
        if r > 1 then
                Tprev := A129392(r-1) ;
        else
                Tprev := [] ;
        fi ;
        T(1,Tprev,r) ;
end:

# return list of the first r elements of row r
A129394 := proc(r) option remember ;
        local Tprev ;
        if r > 1 then
                Tprev := A129394(r-1) ;
        else
                Tprev := [] ;
        fi ;
        T(r-1,Tprev,r) ;
end:

# return list of the first r elements of row r
A129396 := proc(r) option remember ;
```

```
        local Tprev ;
        if r > 1 then
                Tprev := A129396(r-1) ;
        else
                Tprev := [] ;
        fi ;
        T(r,Tprev,r) ;
end:

# return entry r of sequence A129393 (row sum of A129392, r>=1)
A129393 := proc(r)
        local a,i ;
        a:=A129392(r) ; # construct the left part of the row plus diagonal
        op(-1,a)+2*add( op(i,a),i=1..nops(a)-1) ;
end:

# return entry r of sequence A129395 (row sum of A129394, r>=1)
A129395 := proc(r)
        local a,i ;
        a:=A129394(r) ; # construct the left part of the row plus diagonal
        op(-1,a)+2*add( op(i,a),i=1..nops(a)-1) ;
end:

# return element r of sequence A129397 (row sum of A129396, r>=1)
A129397 := proc(r)
        local a,i ;
        a:=A129396(r) ; # construct the left part of the row plus diagonal
        op(-1,a)+2*add( op(i,a),i=1..nops(a)-1) ;
end:

# sample calls
for r from 1 to 10 do
        print( A129392(r)) ;
od ;
for r from 1 to 50 do
        printf("%a, ",A129393(r)) ;
od ;

for r from 1 to 10 do
        print( A129394(r)) ;
od ;
for r from 1 to 50 do
        printf("%a, ",A129395(r)) ;
od ;

for r from 1 to 10 do
        print( A129396(r)) ;
od ;
for r from 1 to 50 do
        printf("%a, ",A129397(r)) ;
od ;
```

[1] N. D. Cahill, J. R. D'Errico, D. A. Narayan, and J. Y. Narayan, Coll. Math. J. **33**, 221 (2002).
[2] R. Bevilacqua, E. Bozzo, G. Temperanza, and P. Zellini, Calcolo **30**, 127 (1993).